
Answers to Chapter 11 Problems

The test scripts assume that the revised Order Entry Database of Chapter 10 is populated according to the text files in the textbook’s website.
1.

SET SERVEROUTPUT ON;

-- Anonymous block to compute the number of days in a non leap year
DECLARE

 NumDays    INTEGER := 0;

 Idx        INTEGER;

BEGIN

-- Use a loop to iterate through the months
 FOR Idx IN 1 .. 12 LOOP
   IF Idx = 1 OR Idx = 3 OR Idx = 5 OR Idx = 7 OR Idx = 8 OR Idx = 10 

              OR Idx = 12 THEN

      NumDays := NumDays + 31;

   ELSIF Idx = 4 OR Idx = 6 OR Idx = 9 OR Idx = 11 THEN

      NumDays := NumDays + 30;

   ELSE

      NumDays := NumDays + 28;

   END IF;

END LOOP;

 -- Display the results

 Dbms_Output.Put_Line('Number of days in a non leap year is ' 
  || To_Char(NumDays) || '.');

END;

/

2. Note that the CASE statement is not supported in Oracle 8i (any version). The solution will not compile in Oracle 8i.
SET SERVEROUTPUT ON;

-- Anonymous block to compute the number of days in a leap year

DECLARE

 NumDays    INTEGER := 0;

 Idx        INTEGER;

BEGIN

-- Use a loop to iterate through the months

 FOR Idx IN 1 .. 12 LOOP

   CASE Idx
     WHEN 1 THEN NumDays := NumDays + 31;

     WHEN 2 THEN NumDays := NumDays + 29;

     WHEN 3 THEN NumDays := NumDays + 31;

     WHEN 4 THEN NumDays := NumDays + 30;

     WHEN 5 THEN NumDays := NumDays + 31;

     WHEN 6 THEN NumDays := NumDays + 30;

     WHEN 7 THEN NumDays := NumDays + 31;

     WHEN 8 THEN NumDays := NumDays + 31;

     WHEN 9 THEN NumDays := NumDays + 30;

     WHEN 10 THEN NumDays := NumDays + 31;

     WHEN 11 THEN NumDays := NumDays + 30;

     WHEN 12 THEN NumDays := NumDays + 31;

  END CASE;

END LOOP;

 -- Display the results

 Dbms_Output.Put_Line('Number of days in a leap year is ' 

  || To_Char(NumDays) || '.');

END;

/

-- Oracle 8i solution using ELSIF instead of CASE

DECLARE

 NumDays    INTEGER := 0;

 Idx        INTEGER;

BEGIN

-- Use a loop to iterate through the months

 FOR Idx IN 1 .. 12 LOOP

   IF Idx = 1 OR Idx = 3 OR Idx = 5 OR Idx = 7 OR Idx = 8 OR Idx = 10 

              OR Idx = 12 THEN

      NumDays := NumDays + 31;

   ELSIF Idx = 4 OR Idx = 6 OR Idx = 9 OR Idx = 11 THEN

      NumDays := NumDays + 30;

   ELSE

      NumDays := NumDays + 29;

   END IF;

END LOOP;

 -- Display the results

 Dbms_Output.Put_Line('Number of days in a non leap year is ' 

  || To_Char(NumDays) || '.');

END;

/

3. 

SET SERVEROUTPUT ON;

-- Anonymous block to compute the future value of 1000 at 8% 

-- for 10 years.
DECLARE

 FutureVal  NUMBER := 1000;

 Idx        INTEGER := 1;

BEGIN

-- Use a loop to iterate through the months

 WHILE Idx <= 10 LOOP

  FutureVal := FutureVal + FutureVal * .08;

  Idx := Idx + 1;

 END LOOP;

 -- Display the results

 Dbms_Output.Put_Line('Future value of 1000 at 8% for 10 years is ' 

  || To_Char(FutureVal) || '.');

END;

/

4. 

SET SERVEROUTPUT ON;

-- Anonymous block to retrieve the product price 

DECLARE

 aProdPrice  Product.ProdPrice%TYPE;
BEGIN

-- Use a loop to iterate through the months

 SELECT ProdPrice INTO aProdPrice FROM Product

  WHERE ProdNo = 'P0036577';
 IF aProdPrice < 100 THEN
    Dbms_Output.Put_Line('Good Buy!' );
 ELSIF aProdPrice BETWEEN 100 and 300 THEN

    Dbms_Output.Put_Line('Competitively Priced!' );

 ELSE

    Dbms_Output.Put_Line('Feature laden product!' );

END IF;

END;

/

5. 

-- Procedure to insert a product and display a message

CREATE OR REPLACE PROCEDURE pr_InsertProductProb5
(aProdNo IN Product.ProdNo%type,

 aProdName IN Product.ProdName%type,

 aProdPrice IN Product.ProdPrice%type,

 aProdQOH IN Product.ProdQOH%type,

 aProdNextShipDate IN Product.ProdNextShipDate%type,

 aSuppNo IN Product.SuppNo%TYPE) IS

BEGIN

INSERT INTO Product 
 (ProdNo, ProdName, ProdPrice, ProdQOH, ProdNextShipDate, SuppNo)

VALUES 
 (aProdNo,aProdName,aProdPrice,aProdQOH,aProdNextShipDate,aSuppNo);

dbms_output.put_line('Added row to the Product table');

EXCEPTION

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Cannot add product row');

END;
/

6. 

-- Procedure to insert a product and generate an output value

CREATE OR REPLACE PROCEDURE pr_InsertProductProb6
(aProdNo IN Product.ProdNo%type,

 aProdName IN Product.ProdName%type,

 aProdPrice IN Product.ProdPrice%type,

 aProdQOH IN Product.ProdQOH%type,

 aProdNextShipDate IN Product.ProdNextShipDate%type,

 aSuppNo IN Product.SuppNo%TYPE, aResult OUT BOOLEAN) IS

BEGIN

INSERT INTO Product 
 (ProdNo, ProdName, ProdPrice, ProdQOH, ProdNextShipDate, SuppNo)

VALUES 
 (aProdNo,aProdName,aProdPrice,aProdQOH,aProdNextShipDate,aSuppNo);

aResult := TRUE;
EXCEPTION

 WHEN Dup_Val_On_Index THEN

   raise_application_error(-20001, 'Primary key not unique');

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Cannot add product row');

END;
/
7. 

SET SERVEROUTPUT ON;

-- Anonymous block to insert procedures 

SELECT COUNT(*) FROM Product;

BEGIN

-- Problem 5 tests

-- This test should succeed.

pr_InsertProductProb5
('P9995688','Battery Back-up',100,12,

  to_date('1-Feb-2004'),'S5095332');

END;

/

SELECT COUNT(*) FROM Product;

ROLLBACK;

-- This test should fail because the primary key is duplicate.

BEGIN

pr_InsertProductProb5

('P9995676','Battery Back-up',100,12,

  to_date('1-Feb-2004'),'S5095332');

END;
SELECT COUNT(*) FROM Product;

ROLLBACK;

-- Problem 6 tests

SELECT COUNT(*) FROM Product;

DECLARE

 Result BOOLEAN;

-- This test should succeed.

BEGIN

pr_InsertProductProb6
('P9995688','Battery Back-up System',100,12,

  to_date('1-Feb-2004'),'S5095332', Result);

IF Result THEN

    dbms_output.put_line('Added a row to the Product table');

ELSE

    dbms_output.put_line('Row not added to the Product table');

END IF;

END;

/

SELECT COUNT(*) FROM Product;

ROLLBACK;

-- This test should fail because the primary key is duplicate.

DECLARE

 Result BOOLEAN;

-- This test should succeed.

BEGIN

pr_InsertProductProb6
('P9995676','Battery Back-up System',100,12,

  to_date('1-Feb-2004'),'S5095332', Result);

IF Result THEN

    dbms_output.put_line('Added a row to the Product table');

ELSE

    dbms_output.put_line('Row not added to the Product table');

END IF;

END;

/

SELECT COUNT(*) FROM Product;

ROLLBACK;

/

8. 

-- Function to determine if the most recent order for a customer
-- was sent to the billing address

CREATE OR REPLACE FUNCTION fn_ShipToBillToMatch
(aCustNo IN Customer.CustNo%type) RETURN BOOLEAN IS

-- Returns TRUE if all address columns match.
-- Returns FALSE if at least one address column does not match.
-- Returns NULL if no customer or no order for the customer.
aCustStreet Customer.CustStreet%type;

aCustCity   Customer.CustCity%type;

aCustState  Customer.CustState%type;

aCustZip    Customer.CustZip%type;

anOrdStreet OrderTbl.OrdStreet%type;

anOrdCity   OrderTbl.OrdCity%type;

anOrdState  OrderTbl.OrdState%type;

anOrdZip    OrderTbl.OrdZip%type;

BEGIN

SELECT OrdStreet, OrdCity, OrdState, OrdZip,
       CustStreet, CustCity, CustState, CustZip
 INTO anOrdStreet, anOrdCity, anOrdState, anOrdZip,
      aCustStreet, aCustCity, aCustState, aCustZip
 FROM OrderTbl, Customer
 WHERE OrderTbl.CustNo = aCustNo
   AND OrderTbl.CustNo = Customer.CustNo

   AND OrdDate = 

   ( SELECT MAX(OrdDate)

      FROM OrderTbl

      WHERE CustNo = aCustNo );
IF anOrdStreet = aCustStreet AND anOrdCity = aCustCity AND
   anOrdState = aCustState AND anOrdZip = aCustZip THEN
   RETURN(TRUE);
ELSE

   RETURN(FALSE);

END IF;
EXCEPTION

 WHEN no_data_found THEN

  RETURN(NULL);

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END;
/

9. There should be 3 test cases for a match, a non match, and a non existing customer.
SET SERVEROUTPUT ON;

BEGIN

-- This test should return TRUE.

IF fn_ShipToBillToMatch('C0954327') THEN
   dbms_output.put_line('Most recent order sent to the 

                         billing address');
ELSE

   dbms_output.put_line('Most recent order not sent to the 

                         billing address');
END IF;

END;

/

-- Modify the ordertbl row so the test will fail.

UPDATE OrderTbl

 SET OrdCity = 'Seattle'

 WHERE OrdNo = 'O8979495';

BEGIN

-- This test should return FALSE.

IF fn_ShipToBillToMatch('C9865874') THEN
   dbms_output.put_line('Most recent order sent to the 

                         billing address');
ELSE

   dbms_output.put_line('Most recent order not sent to the 

                         billing address');
END IF;

END;

/

ROLLBACK;

DECLARE

 FuncResult BOOLEAN;

BEGIN

-- This test should return NULL.

FuncResult := fn_ShipToBillToMatch('C98905074');
IF FuncResult IS NULL THEN
   dbms_output.put_line('No customer found.');
END IF;

END;

/

10. 

CREATE OR REPLACE PROCEDURE pr_ComputeCommission
(anOrdNo IN OrderTbl.OrdNo%type,

 aCommission OUT NUMBER) IS
-- Computes commission for the specified order.

-- If the order does not exist, aCommission is null.
-- If the order does not have an employee, aCommission is 0.

NoEmployee EXCEPTION;

TmpEmpNo OrderTbl.EmpNo%TYPE;

TmpOrdDate OrderTbl.OrdDate%TYPE;

BEGIN

SELECT OrdDate, EmpNo
  INTO tmpOrdDate, tmpEmpNo

  FROM OrderTbl

  WHERE OrdNo = anOrdNo;

IF tmpEmpNo IS NULL THEN

  RAISE NoEmployee;

END IF;

SELECT SUM(EmpCommRate * OrdLine.Qty * ProdPrice) AS Commission

 INTO aCommission

 FROM Employee, OrderTbl, OrdLine, Product

 WHERE OrderTbl.OrdNo = anOrdNo

   AND OrderTbl.OrdNo = OrdLine.OrdNo

   AND OrdLine.ProdNo = Product.ProdNo

   AND OrderTbl.EmpNo = Employee.EmpNo;
EXCEPTION

 WHEN NoEmployee THEN

   aCommission := 0;

 WHEN No_Data_Found THEN
   aCommission := NULL;

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END;
/
11. 

SET SERVEROUTPUT ON;

-- Order with an employee. Order has only 1 order line.
DECLARE
tmpCommission DECIMAL(10,2);

BEGIN

pr_ComputeCommission
('O1116324',tmpCommission);

dbms_output.put_line('Commission is '|| to_char(tmpCommission));

END;

/

-- Order with an employee. Order has 3 order lines.

DECLARE

tmpCommission DECIMAL(10,2);

BEGIN

pr_ComputeCommission

('O1579999',tmpCommission);

dbms_output.put_line('Commission is '|| to_char(tmpCommission));

END;

/

-- Order without an employee. Should generate 0 as the result.
DECLARE

tmpCommission DECIMAL(10,2);

BEGIN

pr_ComputeCommission

('O1241518',tmpCommission);

dbms_output.put_line('Commission is '|| to_char(tmpCommission));

END;

/

-- Non existing order. Should generate null as the result.

DECLARE

tmpCommission DECIMAL(10,2);

BEGIN

pr_ComputeCommission

('O1241599',tmpCommission);

IF tmpCommission IS NULL THEN

   dbms_output.put_line('Order does not exist.');

END IF;

END;

/

12. 

-- Function to determine if the qoh is sufficient for a quantity
-- ordered
CREATE OR REPLACE FUNCTION fn_CheckQOH
(aProdNo IN Product.ProdNo%type, 

 aQuantity IN OrdLine.Qty%TYPE) RETURN BOOLEAN IS

-- Returns TRUE if qoh >= aQuantity
-- Returns FALSE if qoh < aQuantity.
-- Returns NULL if no product exists.

aQOH Product.ProdQOH%TYPE;
BEGIN

SELECT ProdQOH
 INTO aQOH
 FROM Product
 WHERE Product.ProdNo = aProdNo;
IF aQuantity > aQOH THEN
   RETURN(FALSE);
ELSE

   RETURN(TRUE);

END IF;
EXCEPTION

 WHEN no_data_found THEN

  RETURN(NULL);

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END;
/

13. 

-- Procedure to insert a ordline and generate an output value

CREATE OR REPLACE PROCEDURE pr_InsertOrdLine
 (aProdNo IN OrdLine.ProdNo%type,

 anOrdNo IN OrdLine.OrdNo%type,

 aQty IN OrdLine.Qty%type,

 aResult OUT BOOLEAN) IS
 -- aResult is false if insufficient stock

 -- Otherwise aResult is true and insert occurs
OutofStock EXCEPTION;
BEGIN

-- Check for adequate stock
IF NOT(fn_CheckQOH(aProdNo, aQty)) THEN

 RAISE OutofStock;

END IF;

INSERT INTO OrdLine 
 (ProdNo, OrdNo, Qty)

VALUES 
 (aProdNo,anOrdNo,aQty);

aResult := TRUE;
EXCEPTION

 WHEN OutofStock THEN

   aResult := FALSE;

 WHEN Dup_Val_On_Index THEN

   raise_application_error(-20001, 'Primary key not unique');

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Cannot add ordline row');

END;
/
14. 

SET SERVEROUTPUT ON;

-- Test fn_CheckQOH(aProdNo, aQty) directly
SELECT ProdQOH FROM Product WHERE ProdNo = 'P0036566';

-- Test 1: sufficient stock

BEGIN

IF fn_CheckQOH('P0036566', 12) THEN
   dbms_output.put_line('Sufficient Stock');

ELSE

   dbms_output.put_line('Insufficient Stock');

END IF;

END;

/

-- Test 2: insufficient stock

BEGIN

IF fn_CheckQOH('P0036566', 13) THEN

   dbms_output.put_line('Sufficient Stock');

ELSE

   dbms_output.put_line('Insufficient Stock');

END IF;

END;

/

-- Test 3: product does not exist

BEGIN

IF fn_CheckQOH('P0088566', 13) IS NULL THEN

   dbms_output.put_line('Product does not exist');

END IF;

END;

/

-- Test pr_InsertOrdLine

-- Successful insert

DECLARE

aResult BOOLEAN;

BEGIN

pr_InsertOrdLine('P0036566','O1116324',12,aResult);
IF aResult THEN

   dbms_output.put_line('OrdLine insert successful');

ELSE

   dbms_output.put_line(' OrdLine insert not successful');

END IF;

END;

/

ROLLBACK;
-- Failed insert

DECLARE

aResult BOOLEAN;

BEGIN

pr_InsertOrdLine('P0036566','O1116324',13,aResult);

IF aResult THEN

   dbms_output.put_line('OrdLine insert successful');

ELSE

   dbms_output.put_line(' OrdLine insert not successful');

END IF;

END;

/

ROLLBACK;

15. 

CREATE OR REPLACE FUNCTION fn_DetermineMedianBal
 RETURN INTEGER IS

-- Determines the median customer balance.
-- Uses an implicit cursor.

CustCount INTEGER;
MidVal INTEGER;
Idx INTEGER := 1;

MedianVal Customer.CustBal%TYPE;

BEGIN
SELECT COUNT(*) INTO CustCount FROM Customer;

-- Determine middle value

IF Mod(CustCount,2) = 0 THEN

  MidVal := CustCount/2;

ELSE

  MidVal := Trunc(CustCount/2,0) + 1;

END IF;

-- Loop through implicit cursor
FOR CustRec IN 

 ( SELECT CustBal
    FROM Customer
    ORDER BY CustBal ) LOOP

    IF Idx = MidVal THEN

    -- Increment the class rank when the grade changes

       MedianVal := CustRec.CustBal;

       IF Mod(CustCount,2) <> 0 THEN

          RETURN(MedianVal);

       END IF;

    END IF;

    IF Idx = MidVal+1 THEN

       MedianVal := (MedianVal + CustRec.CustBal)/2;

       RETURN(MedianVal);

    END IF;

    Idx := Idx + 1;

END LOOP;

EXCEPTION

WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END;

/

-- Test Script

-- Even number of customers: 16

SELECT COUNT(*) FROM Customer;

SELECT CustBal FROM Customer ORDER BY CustBal;

BEGIN

dbms_output.put_line('Median is '||to_char(fn_DetermineMedianBal));

END;

/

-- Odd number of customers: 17
INSERT INTO Customer(CustNo, CustBal)

   VALUES('11122200', 350);

SELECT COUNT(*) FROM Customer;

SELECT CustBal FROM Customer ORDER BY CustBal;

BEGIN

dbms_output.put_line('Median is '||to_char(fn_DetermineMedianBal));

END;

/

ROLLBACK;
16. 

CREATE OR REPLACE FUNCTION fn_DetermineMedianBalP16
 RETURN INTEGER IS

-- Determines the median customer balance.
-- Uses an explicit cursor.

aCustBal Customer.CustBal%TYPE;
MidVal INTEGER;

CustCount INTEGER;

Idx INTEGER := 1;

MedianVal Customer.CustBal%TYPE;

CURSOR CustCursor IS

 SELECT CustBal

    FROM Customer
    ORDER BY CustBal;
BEGIN
OPEN CustCursor;

-- Determine middle value

SELECT COUNT(*) INTO CustCount FROM Customer;

-- Determine middle value

IF Mod(CustCount,2) = 0 THEN

  MidVal := CustCount/2;

ELSE

  MidVal := Trunc(CustCount/2,0) + 1;

END IF;

-- Loop through explicit cursor

LOOP
    FETCH CustCursor INTO aCustBal;

    IF Idx = MidVal THEN

    -- Increment the class rank when the grade changes

       MedianVal := aCustBal;

       IF Mod(CustCount,2) <> 0 THEN

          CLOSE CustCursor;

          RETURN(MedianVal);

       END IF;

    END IF;

    IF Idx = MidVal+1 THEN

       MedianVal := (MedianVal + aCustBal)/2;

       CLOSE CustCursor;

       RETURN(MedianVal);

    END IF;

    Idx := Idx + 1;

END LOOP;

EXCEPTION

WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END;

/

SET SERVEROUTPUT ON;

-- Test Script

-- Even number of customers: 16

SELECT COUNT(*) FROM Customer;

SELECT CustBal FROM Customer ORDER BY CustBal;

BEGIN

dbms_output.put_line('Median: '||to_char(fn_DetermineMedianBalP16));

END;

/

-- Odd number of customers: 17

INSERT INTO Customer(CustNo, CustBal)

   VALUES('11122200', 350);

SELECT COUNT(*) FROM Customer;

SELECT CustBal FROM Customer ORDER BY CustBal;

BEGIN

dbms_output.put_line('Median:  '||to_char(fn_DetermineMedianBalP16));

END;

/

ROLLBACK;
17. 

CREATE OR REPLACE PACKAGE pck_OrdEntry IS
PROCEDURE pr_InsertProductProb6
(aProdNo IN Product.ProdNo%type,

 aProdName IN Product.ProdName%type,

 aProdPrice IN Product.ProdPrice%type,

 aProdQOH IN Product.ProdQOH%type,

 aProdNextShipDate IN Product.ProdNextShipDate%type,
 aSuppNo IN Product.SuppNo%TYPE, aResult OUT BOOLEAN);

FUNCTION fn_ShipToBillToMatch
(aCustNo IN Customer.CustNo%type) RETURN BOOLEAN;

PROCEDURE pr_ComputeCommission
(anOrdNo IN OrderTbl.OrdNo%type,

 aCommission OUT NUMBER);
PROCEDURE pr_InsertOrdLine

 (aProdNo IN OrdLine.ProdNo%type,

  anOrdNo IN OrdLine.OrdNo%type,

  aQty IN OrdLine.Qty%type,

  aResult OUT BOOLEAN);

FUNCTION fn_DetermineMedianBal RETURN INTEGER;
END pck_OrdEntry;

/

CREATE OR REPLACE PACKAGE BODY pck_OrdEntry IS
FUNCTION fn_CheckQOH
(aProdNo IN Product.ProdNo%type, 

 aQuantity IN OrdLine.Qty%TYPE) RETURN BOOLEAN IS

-- Returns TRUE if qoh >= aQuantity
-- Returns FALSE if qoh < aQuantity.
-- Returns NULL if no product exists.

aQOH Product.ProdQOH%TYPE;
BEGIN

SELECT ProdQOH

 INTO aQOH
 FROM Product
 WHERE Product.ProdNo = aProdNo;

IF aQuantity > aQOH THEN

   RETURN(FALSE);
ELSE

   RETURN(TRUE);

END IF;
EXCEPTION

 WHEN no_data_found THEN

  RETURN(NULL);

WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END fn_CheckQOH;
PROCEDURE pr_InsertProductProb6
(aProdNo IN Product.ProdNo%type,

 aProdName IN Product.ProdName%type,

 aProdPrice IN Product.ProdPrice%type,

 aProdQOH IN Product.ProdQOH%type,

 aProdNextShipDate IN Product.ProdNextShipDate%type,

 aSuppNo IN Product.SuppNo%TYPE, aResult OUT BOOLEAN) IS

BEGIN

INSERT INTO Product 
 (ProdNo, ProdName, ProdPrice, ProdQOH, ProdNextShipDate, SuppNo)

VALUES 
 (aProdNo,aProdName,aProdPrice,aProdQOH,aProdNextShipDate,aSuppNo);

aResult := TRUE;
EXCEPTION

 WHEN Dup_Val_On_Index THEN

   raise_application_error(-20001, 'Primary key not unique');

WHEN OTHERS THEN

   raise_application_error(-20001, 'Cannot add product row');

END pr_InsertProductProb6;
FUNCTION fn_ShipToBillToMatch
(aCustNo IN Customer.CustNo%type) RETURN BOOLEAN IS

-- Returns TRUE if all address columns match.
-- Returns FALSE if at least one address column does not match.
-- Returns NULL if no customer or no order for the customer.
aCustStreet Customer.CustStreet%type;

aCustCity   Customer.CustCity%type;

aCustState  Customer.CustState%type;

aCustZip    Customer.CustZip%type;

anOrdStreet OrderTbl.OrdStreet%type;

anOrdCity   OrderTbl.OrdCity%type;

anOrdState  OrderTbl.OrdState%type;

anOrdZip    OrderTbl.OrdZip%type;

BEGIN

SELECT OrdStreet, OrdCity, OrdState, OrdZip,

       CustStreet, CustCity, CustState, CustZip
 INTO anOrdStreet, anOrdCity, anOrdState, anOrdZip,
      aCustStreet, aCustCity, aCustState, aCustZip
 FROM OrderTbl, Customer
 WHERE OrderTbl.CustNo = aCustNo

   AND OrderTbl.CustNo = Customer.CustNo

   AND OrdDate = 

   ( SELECT MAX(OrdDate)

      FROM OrderTbl

      WHERE CustNo = aCustNo );
IF anOrdStreet = aCustStreet AND anOrdCity = aCustCity AND

   anOrdState = aCustState AND anOrdZip = aCustZip THEN
   RETURN(TRUE);
ELSE

   RETURN(FALSE);

END IF;
EXCEPTION

 WHEN no_data_found THEN

  RETURN(NULL);

WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END fn_ShipToBillToMatch;
PROCEDURE pr_ComputeCommission
(anOrdNo IN OrderTbl.OrdNo%type,

 aCommission OUT NUMBER) IS
-- Computes commission for the specified order.

-- If the order does not exist, aCommission is null.

-- If the order does not have an employee, aCommission is 0.

NoEmployee EXCEPTION;

TmpEmpNo OrderTbl.EmpNo%TYPE;

TmpOrdDate OrderTbl.OrdDate%TYPE;

BEGIN

SELECT OrdDate, EmpNo

  INTO tmpOrdDate, tmpEmpNo

  FROM OrderTbl

  WHERE OrdNo = anOrdNo;

IF tmpEmpNo IS NULL THEN

  RAISE NoEmployee;

END IF;

SELECT SUM(EmpCommRate * OrdLine.Qty * ProdPrice) AS Commission

 INTO aCommission

 FROM Employee, OrderTbl, OrdLine, Product

 WHERE OrderTbl.OrdNo = anOrdNo

   AND OrderTbl.OrdNo = OrdLine.OrdNo

   AND OrdLine.ProdNo = Product.ProdNo

   AND OrderTbl.EmpNo = Employee.EmpNo;

EXCEPTION

 WHEN NoEmployee THEN

   aCommission := 0;

WHEN No_Data_Found THEN

   aCommission := NULL;

WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END pr_ComputeCommission;
PROCEDURE pr_InsertOrdLine

 (aProdNo IN OrdLine.ProdNo%type,

 anOrdNo IN OrdLine.OrdNo%type,

 aQty IN OrdLine.Qty%type,

 aResult OUT BOOLEAN) IS
 -- aResult is false if insufficient stock

 -- Otherwise aResult is true and insert occurs
OutofStock EXCEPTION;
BEGIN

-- Check for adequate stock

IF NOT(fn_CheckQOH(aProdNo, aQty)) THEN

 RAISE OutofStock;

END IF;

INSERT INTO OrdLine 
 (ProdNo, OrdNo, Qty)

VALUES 
 (aProdNo,anOrdNo,aQty);

aResult := TRUE;
EXCEPTION

 WHEN OutofStock THEN

   aResult := FALSE;

 WHEN Dup_Val_On_Index THEN

   raise_application_error(-20001, 'Primary key not unique');

 WHEN OTHERS THEN

   raise_application_error(-20001, 'Cannot add ordline row');

END pr_InsertOrdLine;
FUNCTION fn_DetermineMedianBal
 RETURN INTEGER IS

-- Determines the median customer balance.
-- Uses an implicit cursor.

CustCount INTEGER;
MidVal INTEGER;

Idx INTEGER := 1;

MedianVal Customer.CustBal%TYPE;

BEGIN
SELECT COUNT(*) INTO CustCount FROM Customer;

-- Determine middle value

IF Mod(CustCount,2) = 0 THEN

  MidVal := CustCount/2;

ELSE

  MidVal := Trunc(CustCount/2,0) + 1;

END IF;

-- Loop through implicit cursor
FOR CustRec IN 

 ( SELECT CustBal

    FROM Customer
    ORDER BY CustBal ) LOOP

    IF Idx = MidVal THEN

    -- Increment the class rank when the grade changes

       MedianVal := CustRec.CustBal;

       IF Mod(CustCount,2) <> 0 THEN

          RETURN(MedianVal);

       END IF;

    END IF;

    IF Idx = MidVal+1 THEN

       MedianVal := (MedianVal + CustRec.CustBal)/2;

       RETURN(MedianVal);

    END IF;

    Idx := Idx + 1;

END LOOP;

EXCEPTION

WHEN OTHERS THEN

   raise_application_error(-20001, 'Database error');

END fn_DetermineMedianBal;

END pck_OrdEntry;
SET SERVEROUTPUT ON;

-- Script to test package objects

-- Test pr_InsertProductProb6

DECLARE

Result BOOLEAN;

-- This test should succeed.

BEGIN

pck_OrdEntry.pr_InsertProductProb6

('P9995688','Battery Back-up System',100,12,

  to_date('1-Feb-2004'),'S5095332', Result);

IF Result THEN

    dbms_output.put_line('Added a row to the Product table');

ELSE

    dbms_output.put_line('Row not added to the Product table');

END IF;

END;

/

ROLLBACK;

-- Test fn_ShipToBillToMatch

BEGIN

-- This test should return TRUE.

IF pck_OrdEntry.fn_ShipToBillToMatch('C0954327') THEN
   dbms_output.put_line('Most recent order sent to the 

                         billing address');
ELSE

   dbms_output.put_line('Most recent order not sent to the 

                         billing address');
END IF;

END;

/

-- Test pr_ComputeCommission

-- Order with an employee. Order has only 1 order line.

DECLARE

tmpCommission DECIMAL(10,2);

BEGIN

pck_OrdEntry.pr_ComputeCommission('O1116324',tmpCommission);

dbms_output.put_line('Commission is '|| to_char(tmpCommission));

END;

/

-- Test pr_InsertOrdLine

-- Successful insert

DECLARE

aResult BOOLEAN;

BEGIN

pck_OrdEntry.pr_InsertOrdLine('P0036566','O1116324',12,aResult);

IF aResult THEN

   dbms_output.put_line('OrdLine insert successful');

ELSE

   dbms_output.put_line(' OrdLine insert not successful');

END IF;

END;

/

ROLLBACK;

-- Test fn_DetermineMedianBal
SELECT COUNT(*) FROM Customer;

SELECT CustBal FROM Customer ORDER BY CustBal;

BEGIN

dbms_output.put_line('Median is '||

 to_char(pck_OrdEntry.fn_DetermineMedianBal));

END;

/

18. 

CREATE OR REPLACE TRIGGER tr_Customer_DIUA
AFTER INSERT OR UPDATE OR DELETE

ON Customer
FOR EACH ROW

BEGIN

    dbms_output.put_line('Inserted Table');

    dbms_output.put_line('CustNo: ' || :NEW.CustNo);

    dbms_output.put_line('First Name: '

                      || :NEW.CustFirstName);

    dbms_output.put_line('Last Name: '

                      || :NEW.CustLastName);

    dbms_output.put_line('Street: '

                      || :NEW.CustStreet);

    dbms_output.put_line('City: '

                      || :NEW.CustCity);

    dbms_output.put_line('State: '

                      || :NEW.CustState);

    dbms_output.put_line('Zip: '

                      || :NEW.CustZip);

    dbms_output.put_line('Balance: ' || To_Char(:NEW.CustBal));

    dbms_output.put_line('Deleted Table');

    dbms_output.put_line('CustNo: ' || :OLD.CustNo);

    dbms_output.put_line('First Name: '

                      || :OLD.CustFirstName);

    dbms_output.put_line('Last Name: '

                      || :OLD.CustLastName);

    dbms_output.put_line('Street: '

                      || :OLD.CustStreet);

    dbms_output.put_line('City: '

                      || :OLD.CustCity);

    dbms_output.put_line('State: '

                      || :OLD.CustState);

    dbms_output.put_line('Zip: '

                      || :OLD.CustZip);

    dbms_output.put_line('Balance: ' || To_Char(:OLD.CustBal));

END;
/

SET SERVEROUTPUT ON;

-- Testing script

INSERT INTO Customer (CustNo, CustFirstName, CustLastName, CustBal)

VALUES ('99883457','Joe','Smythe',400);

UPDATE Customer
SET CustBal = 500
WHERE CustNo = '99883457';

DELETE FROM Customer
WHERE CustNo = '99883457';

ROLLBACK;

19. 

CREATE OR REPLACE TRIGGER tr_EmployeeCommRate_UB

-- This trigger ensures that the commission rate does not increase 

-- or decrease by more than 10%.
BEFORE UPDATE OF EmpCommRate
ON Employee
FOR EACH ROW
WHEN  (NEW.EmpCommRate > 1.1 * OLD.EmpCommRate

    OR NEW.EmpCommRate < .9 * OLD.EmpCommRate )
DECLARE 


CommChangeTooLarge EXCEPTION;


ExMessage VARCHAR(200);

BEGIN

    RAISE CommChangeTooLarge;

EXCEPTION

  WHEN CommChangeTooLarge THEN

   -- error number between -20000 and -20999

     ExMessage := 'Commission change exceeds 10%. '; 

     ExMessage := ExMessage || 'Current commission rate: ' || 

              to_char(:OLD.EmpCommRate) || '. ';

     ExMessage := ExMessage || 'New commission rate: ' || 

              to_char(:NEW.EmpCommRate) || '.';

     Raise_Application_Error(-20001, ExMessage);

END;
/

SET SERVEROUTPUT ON;

SELECT EmpNo,EmpCommRate FROM Employee;

-- Test case 1: increase should succeed
UPDATE Employee
SET EmpCommRate = EmpCommRate * 1.1
WHERE EmpNo = 'E1329594';
SELECT EmpNo,EmpCommRate FROM Employee;

-- Test case 2: decrease should succeed

UPDATE Employee
SET EmpCommRate = EmpCommRate * 0.9
WHERE EmpNo = 'E1329594';

SELECT EmpNo,EmpCommRate FROM Employee;

-- Test case 3: increase should fail

UPDATE Employee
SET EmpCommRate = EmpCommRate * 1.2
WHERE EmpNo = 'E1329594';

-- Test case 4: decrease should fail

UPDATE Employee
SET EmpCommRate = EmpCommRate * 0.8
WHERE EmpNo = 'E1329594';

ROLLBACK;

20. Should be BEFORE ROW trigger because updates are made to new values.
CREATE OR REPLACE TRIGGER tr_SupplierURL_IUB

-- This trigger removes the http:// prefix from a web address.

-- The prefix can be made in upper or lower case.

BEFORE INSERT OR UPDATE OF SuppURL
ON Supplier
FOR EACH ROW
DECLARE

 tmpURL VARCHAR2(7);
BEGIN
   tmpURL := SubStr(:NEW.SuppURL,1,7);

   IF Lower(tmpURL) = 'http://' THEN
      :NEW.SuppURL := SubStr(:NEW.SuppURL,8);

   END IF;
END;
/

SET SERVEROUTPUT ON;

-- Test case 1: should remove the URL prefix
INSERT INTO Supplier(SuppNo, SuppName, SuppURL)

VALUES('S9993828','New Supplier','http://www.newsupp.com');

SELECT SuppURL FROM Supplier WHERE SuppNo = 'S9993828';

-- Test case 2: should remove the URL prefix

UPDATE Supplier
SET SuppURL = 'http://www.newsuppbiz.com'
WHERE SuppNo = 'S9993828';

SELECT SuppURL FROM Supplier WHERE SuppNo = 'S9993828';

-- Test case 3: should not change the value

UPDATE Supplier
SET SuppURL = 'www.newsupp.com'
WHERE SuppNo = 'S9993828';

SELECT SuppURL FROM Supplier WHERE SuppNo = 'S9993828';

ROLLBACK;
21.
-- This trigger checks for qoh for both insertion of an order line 
-- and update of ordline.qty.
CREATE OR REPLACE TRIGGER tr_OrdlineQty_IUB
BEFORE INSERT OR UPDATE OF Qty ON OrdLine

FOR EACH ROW

DECLARE my_qoh Product.ProdQOH%TYPE;


out_of_stock exception;


ExMessage VARCHAR(200);

BEGIN

   SELECT ProdQOH
     INTO my_qoh

     FROM Product

     WHERE Product.ProdNo = :NEW.ProdNo;

    IF INSERTING AND (:NEW.qty > my_qoh) THEN

          RAISE out_of_stock;

   END IF;

   IF UPDATING AND (:NEW.qty - :OLD.qty) > my_qoh THEN

          RAISE out_of_stock;

  END IF;

EXCEPTION

  WHEN out_of_stock THEN

   -- error number between -20000 and -20999

     ExMessage := 'Insufficient stock: order quantity of ';

     ExMessage := ExMessage || to_char(:new.qty - :old.qty) || 
                  'is greater than qoh of ';

     ExMessage := ExMessage || to_char(my_qoh) || 
                 ' for product number ' || :new.ProdNo;

     raise_application_error(-20001, ExMessage);

END;
/
22. Should be an AFTER ROW trigger because update propagation.
CREATE OR REPLACE TRIGGER tr_OrdlineQty_DIUA
AFTER INSERT OR DELETE OR UPDATE OF Qty ON OrdLine

FOR EACH ROW

BEGIN

IF UPDATING THEN

  UPDATE Product

    SET ProdQOH = ProdQOH - (:NEW.Qty - :OLD.Qty)

    WHERE Product.ProdNo = :NEW.ProdNo;

END IF;

IF INSERTING THEN

  UPDATE Product

    SET ProdQOH = ProdQOH - :NEW.Qty

    WHERE Product.ProdNo = :NEW.ProdNo;

END IF;

IF DELETING THEN

  UPDATE Product

    SET ProdQOH = ProdQOH + :OLD.Qty

    WHERE Product.ProdNo = :OLD.ProdNo;

END IF;

END;
/
23.

-- Create a new product to use in an order line.

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- This insertion should succeed.
INSERT INTO OrdLine (OrdNo, ProdNo, Qty)

VALUES ('O9919699', 'P0046566', 2);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- This update should fail with an exception raised.

UPDATE OrdLine

  SET Qty = 20

  WHERE OrdNo = 'O9919699' AND ProdNo = 'P0046566';

-- This update should succeed.

UPDATE OrdLine

  SET Qty = 5

  WHERE OrdNo = 'O9919699' AND ProdNo = 'P0046566';

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- This delete should succeed.

DELETE OrdLine

  WHERE OrdNo = 'O9919699' AND ProdNo = 'P0046566';

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
ROLLBACK;
24. Should be an AFTER ROW trigger because update propagation.
CREATE OR REPLACE TRIGGER tr_PurchLine_IA
AFTER INSERT ON PurchLine

FOR EACH ROW

BEGIN

  UPDATE Product

    SET ProdQOH = ProdQOH + :NEW.PurchQty

    WHERE Product.ProdNo = :NEW.ProdNo;

END;
/

-- Create a new product to use in a purchase line.

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- This insertion should succeed.
INSERT INTO PurchLine (PurchNo, ProdNo, PurchQty, PurchUnitCost)

VALUES ('P2224040', 'P0046566', 2, 10);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

ROLLBACK;
25. Should be an AFTER ROW trigger because update propagation.
CREATE OR REPLACE TRIGGER tr_PurchLineQty_UA
AFTER UPDATE OF PurchQty ON PurchLine

FOR EACH ROW

BEGIN

  UPDATE Product

    SET ProdQOH = ProdQOH + ( :NEW.PurchQty - :OLD.PurchQty )
    WHERE Product.ProdNo = :NEW.ProdNo;

END;
/
-- Create a new product to use in a purchase line.

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- Create a new purchase line.
INSERT INTO PurchLine (PurchNo, ProdNo, PurchQty, PurchUnitCost)

VALUES ('P2224040', 'P0046566', 2, 10);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

-- This update should increase the qoh in the related product.
UPDATE PurchLine

 SET PurchQty = 10
 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

-- This update should decrease the qoh in the related product.

UPDATE PurchLine

 SET PurchQty = 5

 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

ROLLBACK;
26. Should be an AFTER ROW trigger because update propagation.
CREATE OR REPLACE TRIGGER tr_PurchLine_DA
AFTER DELETE ON PurchLine

FOR EACH ROW
BEGIN
  UPDATE Product

    SET ProdQOH = ProdQOH - :OLD.PurchQty

    WHERE Product.ProdNo = :OLD.ProdNo;

END;
/
-- Create a new product to use in a purchase line.

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- Create a new purchase line.

INSERT INTO PurchLine (PurchNo, ProdNo, PurchQty, PurchUnitCost)

VALUES ('P2224040', 'P0046566', 2, 10);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

-- Delete the PurchLine. The qoh should decrease.

DELETE PurchLine

 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';
SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

27. Should be an AFTER ROW trigger because update propagation.
CREATE OR REPLACE TRIGGER tr_PurchLineProdNo_UA
AFTER UPDATE OF ProdNo ON PurchLine

FOR EACH ROW
BEGIN

-- Decrease the qoh of the old product
  UPDATE Product

    SET ProdQOH = ProdQOH - :OLD.PurchQty

    WHERE Product.ProdNo = :OLD.ProdNo;

-- Increase the qoh of the new product
  UPDATE Product

    SET ProdQOH = ProdQOH + :NEW.PurchQty

    WHERE Product.ProdNo = :NEW.ProdNo;

END;
/
-- Create two new products to use in a purchase line.

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P1046566','19 inch Color Monitor','S2029929',10,269.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';
-- Create a new purchase line.

INSERT INTO PurchLine (PurchNo, ProdNo, PurchQty, PurchUnitCost)

 VALUES ('P2224040', 'P0046566', 2, 10);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

-- Change the product number

UPDATE PurchLine

 SET ProdNo = 'P1046566'

 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';
-- QOH of old ProdNo should decrease

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- QOH of new ProdNo should increase
SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';

ROLLBACK;
28. The triggers in problems 25 and 27 overlap because they have the same timing, granularity, and applicable table and an UPDATE statement involving both columns causes both triggers to fire. The firing order is arbitrary because the triggers overlap. To see the firing order, display statements have been added to the triggers to see the triggers that fire for the UPDATE statement below. When running the UPDATE statement on Oracle 8.1.7, the ProdNo trigger executes before the PurchQty trigger.
CREATE OR REPLACE TRIGGER tr_PurchLineProdNo_UA
AFTER UPDATE OF ProdNo ON PurchLine

FOR EACH ROW
BEGIN

-- Decrease the qoh of the old product
  dbms_output.put_line('Starting ProdNo Trigger');

  UPDATE Product

    SET ProdQOH = ProdQOH - :OLD.PurchQty

    WHERE Product.ProdNo = :OLD.ProdNo;

-- Increase the qoh of the new product
  UPDATE Product

    SET ProdQOH = ProdQOH + :NEW.PurchQty

    WHERE Product.ProdNo = :NEW.ProdNo;

  dbms_output.put_line('Ending ProdNo Trigger');

END;
/
CREATE OR REPLACE TRIGGER tr_PurchLineQty_UA
AFTER UPDATE OF PurchQty ON PurchLine

FOR EACH ROW

BEGIN

  dbms_output.put_line('Starting PurchQty Trigger');

  UPDATE Product

    SET ProdQOH = ProdQOH + ( :NEW.PurchQty - :OLD.PurchQty )
    WHERE Product.ProdNo = :NEW.ProdNo;

  dbms_output.put_line('Ending PurchQty Trigger');

END;
/
SET SERVEROUTPUT ON;

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P1046566','19 inch Color Monitor','S2029929',10,269.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';
-- Create a new purchase line.

INSERT INTO PurchLine (PurchNo, ProdNo, PurchQty, PurchUnitCost)

 VALUES ('P2224040', 'P0046566', 2, 10);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

-- Change the product number and purchase quantity
UPDATE PurchLine

 SET ProdNo = 'P1046566', PurchQty = 5
 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';

-- QOH of old ProdNo should decrease

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- QOH of new ProdNo should increase

SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';

ROLLBACK;
29. The triggers do not work correctly. The qoh of the new product number is updated twice. The PurchQty trigger should not fire when an UPDATE statement modifies both columns. To handle all three cases (UPDATE statements modifying just one column and UPDATE statements modifying both columns), the two triggers are combined. The combined trigger determines whether one or both columns are updated.
DROP TRIGGER tr_PurchLineProdNo_UA;

DROP TRIGGER tr_PurchLineQty_UA;

CREATE OR REPLACE TRIGGER tr_PurchLineProdNoPurchQty_UA
AFTER UPDATE OF ProdNo, PurchQty ON PurchLine

FOR EACH ROW
BEGIN

-- Decrease the qoh of the old product

IF UPDATING('ProdNo') THEN
  UPDATE Product

    SET ProdQOH = ProdQOH - :OLD.PurchQty

    WHERE Product.ProdNo = :OLD.ProdNo;

-- Increase the qoh of the new product
  UPDATE Product

    SET ProdQOH = ProdQOH + :NEW.PurchQty

    WHERE Product.ProdNo = :NEW.ProdNo;
ELSE

  UPDATE Product

    SET ProdQOH = ProdQOH + ( :NEW.PurchQty - :OLD.PurchQty )
    WHERE Product.ProdNo = :NEW.ProdNo;

END IF;
END;
/
INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P1046566','19 inch Color Monitor','S2029929',10,269.00);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';
-- Create a new purchase line.

INSERT INTO PurchLine (PurchNo, ProdNo, PurchQty, PurchUnitCost)

 VALUES ('P2224040', 'P0046566', 2, 10);

SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

-- Test 1: Change the purchase quantity

UPDATE PurchLine

 SET PurchQty = 5
 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';

-- QOH of ProdNo should increase to 17
SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- Test 2: Change the product number

UPDATE PurchLine

 SET ProdNo = 'P1046566'

 WHERE PurchNo = 'P2224040' AND ProdNo = 'P0046566';

-- QOH of old ProdNo should decrease to 12
SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';
-- QOH of new ProdNo should increase to 15
SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';

-- Test 3: Change the product number and the purchase quantity
UPDATE PurchLine

 SET ProdNo = 'P0046566', PurchQty = 10
 WHERE PurchNo = 'P2224040' AND ProdNo = 'P1046566';

-- QOH of old ProdNo should decrease to 10
SELECT ProdQOH FROM Product WHERE ProdNo = 'P1046566';
-- QOH of new ProdNo should increase to 22
SELECT ProdQOH FROM Product WHERE ProdNo = 'P0046566';

ROLLBACK;
30. An application may choose not to support modification of both columns in an UPDATE statement. To change both the purchase quantity and the product number, an application could require first to delete the previous purchase line, then add a new purchase line with the new product number and quantity. With this requirement, only the purchase quantity trigger would be needed. The product number trigger would not be needed because changing the product number is not supported by the application.
31.

CREATE OR REPLACE TRIGGER tr_ProdPrice_UB
-- This trigger ensures that a price change does not exceed 15%.
-- BEFORE ROW trigger because it enforces a hard constraint.

BEFORE UPDATE OF ProdPrice
ON Product
FOR EACH ROW
WHEN ( NEW.ProdPrice > 1.15 * OLD.ProdPrice OR 

       NEW.ProdPrice < 0.85 * OLD.ProdPrice )
DECLARE 


PriceChangeTooLarge EXCEPTION;


ExMessage VARCHAR(200);

BEGIN

    RAISE PriceChangeTooLarge;

EXCEPTION

  WHEN PriceChangeTooLarge THEN

-- error number between -20000 and -20999

     ExMessage := 'Product price change exceeds 15%. '; 

     ExMessage := ExMessage || 'Current price: ' || 

              to_char(:OLD.ProdPrice) || '. ';

     ExMessage := ExMessage || 'New price: ' || 

              to_char(:NEW.ProdPrice) || '.';

     Raise_Application_Error(-20001, ExMessage);

END;
/

SET SERVEROUTPUT ON;

-- Insert a row for testing
INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';
-- Test case 1: price decrease of 14% should not generate an error.

UPDATE Product

 SET ProdPrice = ProdPrice * 1.14

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

-- Test case 2: price increase of 20% should generate an error.

UPDATE Product

 SET ProdPrice = ProdPrice * 1.20

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

-- Test case 3: price decrease of 20% should generate an error.

UPDATE Product

 SET ProdPrice = ProdPrice * 0.8

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

-- Test case 4: price decrease of 14% should not generate an error.
UPDATE Product

 SET ProdPrice = ProdPrice * 0.86

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

ROLLBACK;

DROP TRIGGER tr_ProdPrice_UB;
32.

-- Create exception table and sequence

CREATE TABLE LogTable
(ExcNo

INTEGER 
 PRIMARY KEY,

 ExcTrigger
VARCHAR2(25) NOT NULL,

 ExcTable

VARCHAR2(25) NOT NULL,

 ExcKeyValue
VARCHAR2(15) NOT NULL,

 ExcDate

DATE DEFAULT SYSDATE NOT NULL,

 ExcText

VARCHAR2(255) NOT NULL );

CREATE SEQUENCE LogSeq INCREMENT BY 1;

CREATE OR REPLACE TRIGGER tr_ProdPrice_UA

-- This trigger inserts a row into LogTable 

-- when a price change exceeds 15%.
-- AFTER ROW trigger because it enforces a soft constraint.

AFTER UPDATE OF ProdPrice
ON Product
FOR EACH ROW
WHEN ( NEW.ProdPrice > 1.15 * OLD.ProdPrice OR 
       NEW.ProdPrice < 0.85 * OLD.ProdPrice )
DECLARE 


PriceChangeTooLarge EXCEPTION;


ExMessage VARCHAR(200);

BEGIN

    RAISE PriceChangeTooLarge;

EXCEPTION

  WHEN PriceChangeTooLarge THEN


INSERT INTO LogTable


 (ExcNo, ExcTrigger, ExcTable, ExcKeyValue, ExcDate, ExcText)


  VALUES (LogSeq.NextVal, 'tr_ProdPrice_UA', 'Product',
              to_char(:New.ProdPrice), SYSDATE, 



     'Price change is more than 15%');
END;
/

SET SERVEROUTPUT ON;

-- Insert a row for testing

INSERT INTO Product (ProdNo, ProdName, SuppNo, ProdQOH, ProdPrice)

 VALUES ('P0046566','17 inch Color Monitor','S2029929',12,169.00);

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';
-- Test case 1: price decrease of 14% should not raise an

-- exception.

UPDATE Product
 SET ProdPrice = ProdPrice * 1.14
 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

SELECT * FROM LogTable;

-- Test case 2: price increase of 20% should raise an exception.

UPDATE Product

 SET ProdPrice = ProdPrice * 1.20

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

SELECT * FROM LogTable;

-- Test case 3: price decrease of 20% should raise an exception.

UPDATE Product

 SET ProdPrice = ProdPrice * 0.8

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

SELECT * FROM LogTable;

-- Test case 4: price decrease of 14% should not raise an

-- exception.

UPDATE Product

 SET ProdPrice = ProdPrice * 0.86

 WHERE ProdNo = 'P0046566';

SELECT ProdPrice FROM Product WHERE ProdNo = 'P0046566';

SELECT * FROM LogTable;

ROLLBACK;

DROP TRIGGER tr_ProdPrice_UB;

DROP TABLE LogTable;

DROP SEQUENCE LogSeq ;
